set scheme s2color

* Figure A5(a)

local nocolumns 5
local nocolumns_m1 = `nocolumns'-1
clear
local new = _N + 1
set obs `new'
gen year = 0
expand `nocolumns_m1'
gen variable = _n+1
tempfile temp0
save `temp0.dta', replace

insheet using "D:\Data\Workdata\703989\FJSJ3989\CIK\replication_restat\output\table_figureA5.txt", tab clear
drop if _n<=3
drop if _n>16
forval i = 2/`nocolumns'{
	replace v`i' = subinstr(v`i',"*","",.)
	replace v`i' = subinstr(v`i',"(","",.)
	replace v`i' = subinstr(v`i',")","",.)
}
destring v2-v`nocolumns', replace
gen coefficient = (v1!="")
carryforward v1, replace
tempfile temp1
save `temp1.dta', replace

keep if coefficient==1
rename v1 time
reshape long v, i(time) j(variable)
drop coefficient
rename v coefficient
tempfile temp2
save `temp2.dta', replace

use `temp1.dta', clear
keep if coefficient==0
rename v1 time
reshape long v, i(time) j(variable)
drop coefficient
rename v stderr
merge 1:1 time variable using `temp2.dta', nogen
gen lb = coefficient - 1.96*stderr
gen ub = coefficient + 1.96*stderr

gen year = .
forval i = 1/4{
	replace year = -`i' if time =="m`i'"
	replace year = `i' if time =="p`i'"
}
append using `temp0.dta'
replace coefficient = 0 if year==0
replace lb = 0 if year==0
replace ub = 0 if year==0

drop if year==-4 | year==4

replace year = year - 0.0625
gen nudge2 = year + 0.125

sort variable year
graph twoway (scatter coefficient year if variable==2, msymbol(circle) mcolor(navy)) (line coefficient year if variable==2, lcolor(navy) lwidth(medthick)) (rcap ub lb year if variable==2, lcolor(navy) lwidth(medthick)) (scatter coefficient nudge2 if variable==3, msymbol(diamond) mcolor(cranberry) yaxis(2)) (line coefficient nudge2 if variable==3, lcolor(cranberry) lpattern(longdash) lwidth(medthick) yaxis(2)) (rcap ub lb nudge2 if variable==3, lcolor(cranberry) lwidth(medthick) yaxis(2)) , graphregion(color(white)) xline(0,lwidth(3) lcolor(gs14) lstyle(solid)) ytitle("Coefficient estimate" "", size(large)) xtitle("Year relative to merger", size(large)) ylabel(-0.2(0.1)0.2, labsize(large) gmax gmin) ylabel(-0.6(0.2)0.6, labsize(large) axis(2) ) xlabel(-3 "-3" -2 "-2" -1 "-1" 0 "0" 1 "+1" 2 "+2" 3 "+3", labsize(large)) legend(order(1 "Prod{superscript:D} (LHS)" 4 "Prod{superscript:D,AT} (RHS)") size(large) pos(6) row(1) region(lcolor(black))) title("Poisson" " ", size(vlarge) color(black))
graph export "D:\Data\Workdata\703989\FJSJ3989\CIK\replication_restat\output\figure_A5a.pdf", replace mag(100)

********************************************************************************

* Figure A5(b)

sort variable year
graph twoway (scatter coefficient year if variable==4, msymbol(circle) mcolor(navy)) (line coefficient year if variable==4, lcolor(navy) lwidth(medthick)) (rcap ub lb year if variable==4, lcolor(navy) lwidth(medthick)) (scatter coefficient nudge2 if variable==5, msymbol(diamond) mcolor(cranberry)) (line coefficient nudge2 if variable==5, lcolor(cranberry) lpattern(longdash) lwidth(medthick)) (rcap ub lb nudge2 if variable==5, lcolor(cranberry) lwidth(medthick)) , graphregion(color(white)) xline(0,lwidth(3) lcolor(gs14) lstyle(solid)) ytitle("Coefficient estimate" "", size(large)) xtitle("Year relative to merger", size(large)) ylabel(,labsize(large)) xlabel(-3 "-3" -2 "-2" -1 "-1" 0 "0" 1 "+1" 2 "+2" 3 "+3", labsize(large)) legend(order(1 "Mkt{superscript:X}" 4 "Prod{superscript:X}") size(large) pos(6) row(1) region(lcolor(black))) title("Poisson" " ", size(vlarge) color(black))
graph export "D:\Data\Workdata\703989\FJSJ3989\CIK\replication_restat\output\figure_A5b.pdf", replace mag(100)
